F_{35} の導出 with Mathematica14.0

In[3273]:=
```
ClearAll["`*"]
```

§1 F_{35} と A_i の定義

既約で有理数係数かつmonicな7次方程式
「$f = x^7 + a_1 x^6 + a_2 x^5 + a_3 x^4 + a_4 x^3 + a_5 x^2 + a_6 x + a_7 = 0$」の解を x_0, x_1, ..., x_6 とします．$x_0 \sim x_6$ から異なる3つの解x_i, x_j, x_k $(0 \le i < j < k \le 6)$ を選んでその和s_{ijk} を作ります．この時，
fの補助方程式F_{35} とその係数 A_i を次の様に定義します．

$$F_{35} = (x - s_{012})(x - s_{013}) \cdots (x - s_{456}) = \prod_{0\le i<j<k\le 6} (x - s_{ijk}) = x^{35} + A_1 x^{34} + A_2 x^{33} + \cdots + A_{34} x + A_{35}$$

F_{35} は ${}_7C_3 = 35$ 次で $x_0 \sim x_6$ の対称式なので その $(35 - i)$ 次の係数を A_i とすると A_i は有理数です．例えば A_1, A_2 , A_3 は以下の様に求まります．

In[3274]:=
```
vars = {x0, x1, x2, x3, x4, x5, x6};
s = Total /@ Subsets[vars, {3}] (*=s_{i,j,k}*)
```

Out[3275]=
```
{x0 + x1 + x2, x0 + x1 + x3, x0 + x1 + x4, x0 + x1 + x5, x0 + x1 + x6, x0 + x2 + x3, x0 + x2 + x4,
 x0 + x2 + x5, x0 + x2 + x6, x0 + x3 + x4, x0 + x3 + x5, x0 + x3 + x6, x0 + x4 + x5, x0 + x4 + x6,
 x0 + x5 + x6, x1 + x2 + x3, x1 + x2 + x4, x1 + x2 + x5, x1 + x2 + x6, x1 + x3 + x4, x1 + x3 + x5,
 x1 + x3 + x6, x1 + x4 + x5, x1 + x4 + x6, x1 + x5 + x6, x2 + x3 + x4, x2 + x3 + x5, x2 + x3 + x6,
 x2 + x4 + x5, x2 + x4 + x6, x2 + x5 + x6, x3 + x4 + x5, x3 + x4 + x6, x3 + x5 + x6, x4 + x5 + x6}
```

In[3276]:=
```
$F35[x_] = Product[x - s[[i]], {i, 1, 35}]
```

Out[3276]=
```
(x - x0 - x1 - x2) (x - x0 - x1 - x3) (x - x0 - x2 - x3) (x - x1 - x2 - x3) (x - x0 - x1 - x4)
 (x - x0 - x2 - x4) (x - x1 - x2 - x4) (x - x0 - x3 - x4) (x - x1 - x3 - x4) (x - x2 - x3 - x4)
 (x - x0 - x1 - x5) (x - x0 - x2 - x5) (x - x1 - x2 - x5) (x - x0 - x3 - x5) (x - x1 - x3 - x5)
 (x - x2 - x3 - x5) (x - x0 - x4 - x5) (x - x1 - x4 - x5) (x - x2 - x4 - x5) (x - x3 - x4 - x5)
 (x - x0 - x1 - x6) (x - x0 - x2 - x6) (x - x1 - x2 - x6) (x - x0 - x3 - x6) (x - x1 - x3 - x6)
 (x - x2 - x3 - x6) (x - x0 - x4 - x6) (x - x1 - x4 - x6) (x - x2 - x4 - x6) (x - x3 - x4 - x6)
 (x - x0 - x5 - x6) (x - x1 - x5 - x6) (x - x2 - x5 - x6) (x - x3 - x5 - x6) (x - x4 - x5 - x6)
```

(後でも F35を定義するので，こちらは先頭に「$」を付加しました．以下，適時，先頭に「$や別の文字」を付加します．)

SymmetricReduction で A1,A2,A3を a1~a7 の式に直します．

In[∘]:=

SymmetricReduction[f, {x_1, …, x_n}]
x_1, …, x_nにおいて $f == p + q$ であるような多項式のペア{p, q}を与える．ただし，p は対称な部分であり q は剰余である.

SymmetricReduction [f, {x_1, …, x_n}, {s_1, …, s_n}]
p における初等対称式を s_1, …, s_n で置き換えたペア {p, q} を与える.

【例】SymmetricReduction[x^2 - y^2, {x, y}, {s1, s2}] → $\{s1^2 - 2\,s2,\ -2\,y^2\}$

In[3277]:=

```
AA[m_] := (-1)^m * Total[(Times @@ # &) /@ Subsets[s, {m}]]
```

In[3278]:=

```
A1 = SymmetricReduction[AA[1], vars, {-a1, a2, -a3, a4, -a5, a6, -a7}][[1]]
A2 = SymmetricReduction[AA[2], vars, {-a1, a2, -a3, a4, -a5, a6, -a7}][[1]]
A3 = SymmetricReduction[AA[3], vars, {-a1, a2, -a3, a4, -a5, a6, -a7}][[1]]
```

Out[3278]=

15 a1

Out[3279]=

$105\,a1^2 + 10\,a2$

Out[3280]=

$455\,a1^3 + 140\,a1\,a2 + 2\,a3$

しかしこの方法では A_{20} などになると計算量が多すぎて不可能です.
原論文ではこの課題を ニュートン恒等式 (ニュートン和) を用いて見事に解決しています.
ニュートン恒等式については例えば https://manabitimes.jp/math/1304 をご覧ください.

【コメント】例えば A2の式をすべて挙げたものは下の closed cell に入っています.

§2 S_i, $S_{i,j}$, $S_{i,j,k}$,T[m] の定義

§2-1 以下の定義では fの次数nは7とは限定せず n≥ 3 とします.

$$\begin{cases} f = 0 \text{ の異なる解 の } i \text{ 乗の総和を } S_i \ (1 \le i) \\ f = 0 \text{ の異なる解を } x, y \text{ としたとき } x^i y^j \text{ の総和を } S_{i,j} \ (1 \le i \le j) \\ f = 0 \text{ の異なる解を } x, y, z \text{ としたとき } x^i y^j z^k \text{ の総和を } S_{i,j,k} \ (1 \le i \le j \le k) \end{cases}$$

定義より「$S_{i,j}=S_{j,i}$， $S_{i,j,k}=S_{i,k,j}=S_{j,i,k}=S_{j,k,i}=S_{k,i,j}=S_{k,j,i}$ 」です.

Mathematica では n=7のとき， 次の様に定義できます.

In[3282]:=

```
$S[i_] := Total[vars^i]
u[{a_, b_}, {i_, j_}] := a^i b^j + a^j b^i
u[{a_, b_, c_}, {i_, j_, k_}] :=
 a^i b^j c^k + a^i b^k c^j + a^j b^i c^k + a^j b^k c^i + a^k b^i c^j + a^k b^j c^i
$S[i_, j_] :=
 u[#, {i, j}] & /@ (Subsets[vars, {2}]) / Factorial[Max[Tally[{i, j}][[All, 2]]]] // Total
$S[i_, j_, k_] := u[#, {i, j, k}] & /@ (Subsets[vars, {3}]) /
    Factorial[Max[Tally[{i, j, k}][[All, 2]]]] // Total // Expand
```

In[3287]:=

```
$S[2] (*=S2*)
$S[2, 3] (*=S2,3*)
$S[2, 3, 4] (*=S2,3,4*)
```

Out[3287]=

$x0^2 + x1^2 + x2^2 + x3^2 + x4^2 + x5^2 + x6^2$

Out[3288]=

$x0^3 x1^2 + x0^2 x1^3 + x0^3 x2^2 + x1^3 x2^2 + x0^2 x2^3 + x1^2 x2^3 + x0^3 x3^2 + x1^3 x3^2 + x2^3 x3^2 + x0^2 x3^3 + x1^2 x3^3 +$
$x2^2 x3^3 + x0^3 x4^2 + x1^3 x4^2 + x2^3 x4^2 + x3^3 x4^2 + x0^2 x4^3 + x1^2 x4^3 + x2^2 x4^3 + x3^2 x4^3 + x0^3 x5^2 +$
$x1^3 x5^2 + x2^3 x5^2 + x3^3 x5^2 + x4^3 x5^2 + x0^2 x5^3 + x1^2 x5^3 + x2^2 x5^3 + x3^2 x5^3 + x4^2 x5^3 + x0^3 x6^2 +$
$x1^3 x6^2 + x2^3 x6^2 + x3^3 x6^2 + x4^3 x6^2 + x5^3 x6^2 + x0^2 x6^3 + x1^2 x6^3 + x2^2 x6^3 + x3^2 x6^3 + x4^2 x6^3 + x5^2 x6^3$

Out[3289]=

$x0^4 x1^3 x2^2 + x0^3 x1^4 x2^2 + x0^4 x1^2 x2^3 + x0^2 x1^4 x2^3 + x0^3 x1^2 x2^4 + x0^2 x1^3 x2^4 + x0^4 x1^3 x3^2 + x0^3 x1^4 x3^2 + x0^4 x2^3 x3^2 +$
$x1^4 x2^3 x3^2 + x0^3 x2^4 x3^2 + x1^3 x2^4 x3^2 + x0^4 x1^2 x3^3 + x0^2 x1^4 x3^3 + x0^4 x2^2 x3^3 + x1^4 x2^2 x3^3 + x0^2 x2^4 x3^3 + x1^2 x2^4 x3^3 +$
$x0^3 x1^2 x3^4 + x0^2 x1^3 x3^4 + x0^3 x2^2 x3^4 + x1^3 x2^2 x3^4 + x0^2 x2^3 x3^4 + x1^2 x2^3 x3^4 + x0^4 x1^3 x4^2 + x0^3 x1^4 x4^2 + x0^4 x2^3 x4^2 +$
$x1^4 x2^3 x4^2 + x0^3 x2^4 x4^2 + x1^3 x2^4 x4^2 + x0^4 x3^3 x4^2 + x1^4 x3^3 x4^2 + x2^4 x3^3 x4^2 + x0^3 x3^4 x4^2 + x1^3 x3^4 x4^2 + x2^3 x3^4 x4^2 +$
$x0^4 x1^2 x4^3 + x0^2 x1^4 x4^3 + x0^4 x2^2 x4^3 + x1^4 x2^2 x4^3 + x0^2 x2^4 x4^3 + x1^2 x2^4 x4^3 + x0^4 x3^2 x4^3 + x1^4 x3^2 x4^3 + x2^4 x3^2 x4^3 +$
$x0^2 x3^4 x4^3 + x1^2 x3^4 x4^3 + x2^2 x3^4 x4^3 + x0^3 x1^2 x4^4 + x0^2 x1^3 x4^4 + x0^3 x2^2 x4^4 + x1^3 x2^2 x4^4 + x0^2 x2^3 x4^4 + x1^2 x2^3 x4^4 +$
$x0^3 x3^2 x4^4 + x1^3 x3^2 x4^4 + x2^3 x3^2 x4^4 + x0^2 x3^3 x4^4 + x1^2 x3^3 x4^4 + x2^2 x3^3 x4^4 + x0^4 x1^3 x5^2 + x0^3 x1^4 x5^2 + x0^4 x2^3 x5^2 +$
$x1^4 x2^3 x5^2 + x0^3 x2^4 x5^2 + x1^3 x2^4 x5^2 + x0^4 x3^3 x5^2 + x1^4 x3^3 x5^2 + x2^4 x3^3 x5^2 + x0^3 x3^4 x5^2 + x1^3 x3^4 x5^2 + x2^3 x3^4 x5^2 +$
$x0^4 x4^3 x5^2 + x1^4 x4^3 x5^2 + x2^4 x4^3 x5^2 + x3^4 x4^3 x5^2 + x0^3 x4^4 x5^2 + x1^3 x4^4 x5^2 + x2^3 x4^4 x5^2 + x3^3 x4^4 x5^2 + x0^4 x1^2 x5^3 +$
$x0^2 x1^4 x5^3 + x0^4 x2^2 x5^3 + x1^4 x2^2 x5^3 + x0^2 x2^4 x5^3 + x1^2 x2^4 x5^3 + x0^4 x3^2 x5^3 + x1^4 x3^2 x5^3 + x2^4 x3^2 x5^3 + x0^2 x3^4 x5^3 +$
$x1^2 x3^4 x5^3 + x2^2 x3^4 x5^3 + x0^4 x4^2 x5^3 + x1^4 x4^2 x5^3 + x2^4 x4^2 x5^3 + x3^4 x4^2 x5^3 + x0^2 x4^4 x5^3 + x1^2 x4^4 x5^3 + x2^2 x4^4 x5^3 +$
$x3^2 x4^4 x5^3 + x0^3 x1^2 x5^4 + x0^2 x1^3 x5^4 + x0^3 x2^2 x5^4 + x1^3 x2^2 x5^4 + x0^2 x2^3 x5^4 + x1^2 x2^3 x5^4 + x0^3 x3^2 x5^4 + x1^3 x3^2 x5^4 +$
$x2^3 x3^2 x5^4 + x0^2 x3^3 x5^4 + x1^2 x3^3 x5^4 + x2^2 x3^3 x5^4 + x0^3 x4^2 x5^4 + x1^3 x4^2 x5^4 + x2^3 x4^2 x5^4 + x3^3 x4^2 x5^4 + x0^2 x4^3 x5^4 +$
$x1^2 x4^3 x5^4 + x2^2 x4^3 x5^4 + x3^2 x4^3 x5^4 + x0^4 x1^3 x6^2 + x0^3 x1^4 x6^2 + x0^4 x2^3 x6^2 + x1^4 x2^3 x6^2 + x0^3 x2^4 x6^2 + x1^3 x2^4 x6^2 +$
$x0^4 x3^3 x6^2 + x1^4 x3^3 x6^2 + x2^4 x3^3 x6^2 + x0^3 x3^4 x6^2 + x1^3 x3^4 x6^2 + x2^3 x3^4 x6^2 + x0^4 x4^3 x6^2 + x1^4 x4^3 x6^2 + x2^4 x4^3 x6^2 +$
$x3^4 x4^3 x6^2 + x0^3 x4^4 x6^2 + x1^3 x4^4 x6^2 + x2^3 x4^4 x6^2 + x3^3 x4^4 x6^2 + x0^4 x5^3 x6^2 + x1^4 x5^3 x6^2 + x2^4 x5^3 x6^2 + x3^4 x5^3 x6^2 +$
$x4^4 x5^3 x6^2 + x0^3 x5^4 x6^2 + x1^3 x5^4 x6^2 + x2^3 x5^4 x6^2 + x3^3 x5^4 x6^2 + x4^3 x5^4 x6^2 + x0^4 x1^2 x6^3 + x0^2 x1^4 x6^3 + x0^4 x2^2 x6^3 +$
$x1^4 x2^2 x6^3 + x0^2 x2^4 x6^3 + x1^2 x2^4 x6^3 + x0^4 x3^2 x6^3 + x1^4 x3^2 x6^3 + x2^4 x3^2 x6^3 + x0^2 x3^4 x6^3 + x1^2 x3^4 x6^3 +$
$x2^2 x3^4 x6^3 + x0^4 x4^2 x6^3 + x1^4 x4^2 x6^3 + x2^4 x4^2 x6^3 + x3^4 x4^2 x6^3 + x0^2 x4^4 x6^3 + x1^2 x4^4 x6^3 + x2^2 x4^4 x6^3 +$
$x3^2 x4^4 x6^3 + x0^4 x5^2 x6^3 + x1^4 x5^2 x6^3 + x2^4 x5^2 x6^3 + x3^4 x5^2 x6^3 + x4^4 x5^2 x6^3 + x0^2 x5^4 x6^3 + x1^2 x5^4 x6^3 +$
$x2^2 x5^4 x6^3 + x3^2 x5^4 x6^3 + x4^2 x5^4 x6^3 + x0^3 x1^2 x6^4 + x0^2 x1^3 x6^4 + x0^3 x2^2 x6^4 + x1^3 x2^2 x6^4 + x0^2 x2^3 x6^4 +$
$x1^2 x2^3 x6^4 + x0^3 x3^2 x6^4 + x1^3 x3^2 x6^4 + x2^3 x3^2 x6^4 + x0^2 x3^3 x6^4 + x1^2 x3^3 x6^4 + x2^2 x3^3 x6^4 + x0^3 x4^2 x6^4 +$
$x1^3 x4^2 x6^4 + x2^3 x4^2 x6^4 + x3^3 x4^2 x6^4 + x0^2 x4^3 x6^4 + x1^2 x4^3 x6^4 + x2^2 x4^3 x6^4 + x3^2 x4^3 x6^4 + x0^3 x5^2 x6^4 +$
$x1^3 x5^2 x6^4 + x2^3 x5^2 x6^4 + x3^3 x5^2 x6^4 + x4^3 x5^2 x6^4 + x0^2 x5^3 x6^4 + x1^2 x5^3 x6^4 + x2^2 x5^3 x6^4 + x3^2 x5^3 x6^4 + x4^2 x5^3 x6^4$

§2-2 ニュートン恒等式(S_i と a_i の関係式)

n=7の時，ニュートン恒等式は次の様になります．

$$\begin{cases} -S_1 = a_1 & \square \\ -S_i = a_1 S_{i-1} + a_2 S_{i-2} + \cdots + a_{i-1} S_1 + i\, a_i & (2 \le i \le 7) \\ -S_i = a_1 S_{i-1} + a_2 S_{i-2} + \cdots + a_6 S_{i-6} + a_7 S_{i-7} & (i > 7) \end{cases}$$

【コメント】2 次方程式については i>n の式は大学入試でも頻出です．ここでは3 次方程式 $x^3 + a_1 x^2 + a_2 x + a_3 = 0$ の解を α, β, γ とします．このとき

$$\begin{cases} \alpha^3 + a_1 \alpha^2 + a_2 \alpha + a_3 = 0 & (ア) \\ \beta^3 + a_1 \beta^2 + a_2 \beta + a_3 = 0 & (イ) \\ \gamma^3 + a_1 \gamma^2 + a_2 \gamma + a_3 = 0 & (ウ) \end{cases}$$

$n = 1, 2 ..$ の時「$(ア) * \alpha^n + (イ) * \beta^n + (ウ) * \gamma^n$」より

$$\begin{cases} \alpha^{n+3} + a_1 \alpha^{n+2} + a_2 \alpha^{n+1} + a_3 \alpha^n = 0 \\ \beta^{n+3} + a_1 \beta^{n+2} + a_2 \beta^{n+1} + a_3 \beta^n = 0 \\ \gamma^{n+3} + a_1 \gamma^{n+2} + a_2 \gamma^{n+1} + a_3 \gamma^n = 0 \end{cases}$$

両辺を加えて

$$S_{n+3} + a_1 S_{n+2} + a_2 S_{n+1} + a_3 S_n = 0$$

即ち

$$-S_i = a_1 S_{i-1} + a_2 S_{i-2} + a_3 S_{i-3} \ (i = 4, 5, 6 ..)$$

$1 \le i \le 3$ に対しては 次の様になります.

$$\begin{cases} -S_3 = a_1 S_2 + a_2 S_1 + 3 a_3 \ (\text{この証明は上と同じ}) & (\text{エ}) \\ -S_2 = a_1 S_1 + 2 a_2 \ (\text{これの一般的証明はちょっと難しい}) & (\text{オ}) \\ -S_1 = a_1 \ (\text{解と係数の関係}) & (\text{カ}) \end{cases}$$

§2-3 S_i, $s_{i,j}$, $S_{i,j,k}$ の関係

次に $s_{i,j}$ と s_i の関係式は次の様になります.

In[ ]:=

$$\begin{cases} s_{i,j} = s_i s_j - s_{i+j} & (i \neq j) \\ s_{i,i} = \left(s_i^2 - s_{2i}\right) / 2 & (i = j) \end{cases}$$

例えば 3 次方程式の時は, 下より「 $S_{2,3}$=$S_2 S_3 - S_5$, $S_{3,3}$= ($\mathbf{S_3^2 - S_6}$)/2」 などが確認できます.

$$\begin{cases} S_{2,3} = \alpha^2 \beta^3 + \alpha^3 \beta^2 + \beta^2 \gamma^3 + \beta^3 \gamma^2 + \gamma^2 \alpha^3 + \gamma^3 \alpha^2 & \square \\ S_2 S_3 - S_5 = (\alpha^2 + \beta^2 + \gamma^2)(\alpha^3 + \beta^3 + \gamma^3) - (\alpha^5 + \beta^5 + \gamma^5) & \square \\ S_{3,3} = \alpha^3 \beta^3 + \beta^3 \gamma^3 + \gamma^3 \alpha^3 & \square \\ S_3^2 - S_6 = (\alpha^3 + \beta^3 + \gamma^3)^2 - (\alpha^6 + \beta^6 + \gamma^6) & \square \end{cases}$$

$s_{i,j,k}$ を $s_{i,j}$ と s_i で表す式は次の様になります.

$$\begin{cases} s_{i,j,k} = s_i s_{j,k} - s_{i+j,k} - s_{i+k,j} & (i, j, k \text{ が全て異なるとき}) \\ s_{i,i,k} = s_k s_{i,i} - s_{i+k,i} & (i = j \neq k) \\ s_{i,k,k} = s_i s_{k,k} - s_{i+k,k} & (i \neq j = k) \\ s_{i,i,i} = (s_i s_{i,i} - s_{2i,i}) / 3 & (i = j = k) \end{cases}$$

例えば 3 次方程式の時は, 下より「$S_{2,3,4}$= $S_2 S_{3,4}$ −$S_{5,4}$−$S_{6,3}$, $S_{3,3,2}$= $S_2 S_{3,3}$ −$S_{5,3}$, $S_{3,3,3}$= ($S_3 S_{3,3}$ −$S_{6,3}$)/3」 などが確認できます.

$S_{2,3,4} = \alpha^2\beta^3\gamma^4 + \alpha^2\beta^4\gamma^3 + \alpha^3\beta^2\gamma^4 + \alpha^3\beta^4\gamma^2 + \alpha^4\beta^2\gamma^3 + \alpha^4\beta^3\gamma^2$

$S_2 S_{3,4} - S_{5,4} - S_{6,3} = (\alpha^2+\beta^2+\gamma^2)(\alpha^3\beta^4 + \alpha^4\beta^3 + \beta^3\gamma^4 + \beta^4\gamma^3 + \gamma^3\alpha^4 + \gamma^4\alpha^3) - (\alpha^5\beta^4 + \alpha^4\beta^5 + \beta^5\gamma^4 + \beta^4\gamma^5 + \gamma^5\alpha^4 + \gamma^4\alpha^5) - (\alpha^6\beta^3 + \alpha^3\beta^6 + \beta^6\gamma^3 + \beta^3\gamma^6 + \gamma^6\alpha^3 + \gamma^3\alpha^6)$

$S_{3,3,2} = \alpha^3\beta^3\gamma^2 + \alpha^3\beta^2\gamma^3 + \alpha^2\beta^3\gamma^3$

$S_2 S_{3,3} - S_{5,3} = (\alpha^2+\beta^2+\gamma^2)(\alpha^3\beta^3 + \beta^3\gamma^3 + \gamma^3\alpha^3) - (\alpha^5\beta^3 + \alpha^3\beta^5 + \beta^5\gamma^3 + \beta^3\gamma^5 + \gamma^5\alpha^3 + \gamma^3\alpha^5)$

$S_{3,3,3} = \alpha^3\beta^3\gamma^3$

$S_3 S_{3,3} - S_{6,3} = (\alpha^3+\beta^3+\gamma^3)(\alpha^3\beta^3 + \beta^3\gamma^3 + \gamma^3\alpha^3) - (\alpha^6\beta^3 + \alpha^3\beta^6 + \beta^6\gamma^3 + \beta^3\gamma^6 + \gamma^6\alpha^3 + \gamma^3\alpha^6)$

§2-4 T[m]の定義と S_i, $s_{i,j}$, $S_{i,j,k}$との関係

これら S_i, $S_{i,j}$, $S_{i,j,k}$ を用いて「$T[m] = \sum_{x,y,z\text{はfの異なる解}} (x+y+z)^m$ 」が求められます.

ただし, $m \geq 1$ で,

Σは「f の解から異なる3つの解 x, y, z を取る全てのケースを考え, 和を取る」という意味です.

例えば 7 次方程式の場合はT[5] は次の様になります.

In[3290]:=

```
$T[m_] := Total[Power[#, m] & /@ Total /@ Subsets[vars, {3}]]
$T[5]
```

Out[3291]=

$(x0+x1+x2)^5 + (x0+x1+x3)^5 + (x0+x2+x3)^5 + (x1+x2+x3)^5 + (x0+x1+x4)^5 + (x0+x2+x4)^5 + (x1+x2+x4)^5 + (x0+x3+x4)^5 + (x1+x3+x4)^5 + (x2+x3+x4)^5 + (x0+x1+x5)^5 + (x0+x2+x5)^5 + (x1+x2+x5)^5 + (x0+x3+x5)^5 + (x1+x3+x5)^5 + (x2+x3+x5)^5 + (x0+x4+x5)^5 + (x1+x4+x5)^5 + (x2+x4+x5)^5 + (x3+x4+x5)^5 + (x0+x1+x6)^5 + (x0+x2+x6)^5 + (x1+x2+x6)^5 + (x0+x3+x6)^5 + (x1+x3+x6)^5 + (x2+x3+x6)^5 + (x0+x4+x6)^5 + (x1+x4+x6)^5 + (x2+x4+x6)^5 + (x3+x4+x6)^5 + (x0+x5+x6)^5 + (x1+x5+x6)^5 + (x2+x5+x6)^5 + (x3+x5+x6)^5 + (x4+x5+x6)^5$

以下 $m \geq 1$, i, j,

$k \geq 0$ に対して「$\binom{m}{i} = {}_mC_i$, $\binom{m}{i,\ j,\ k} = \dfrac{m!}{i!\ j!\ k!}$ (但し $i+j+k=m$)」とします.

上のT[5] を展開したときの同類項の個数を考えます.

上の例で $x0^5$ の個数は $(x0+xj+xk)^5$ (0, j, kはすべて異なる「xの添え数字」)

の個数と等しいから $\binom{6}{2}$ (j, kを選ぶ組み合わせ ${}_6C_2$) = 15 です.

次に $x0^2\ x1^3$ の個数を求めます. $(x0+x1+x2)^5$ を展開したときの $x0^2\ x1^3$ の個数は,

多項定理より $5!/(2!\times 3!) = \binom{5}{2}$ です.

$(x0+x1+xk)^5$ (0, 1, kはすべて異なる「xの添え数字」) の個数は $\binom{5}{1} = {}_5C_1$ なので,

T[5] の $x0^2\ x1^3$ の個数は $5 * \binom{5}{2}$ です.

更に $x0^2\ x1^2\ x2^1$ の個数を求めます.

$(x0+x1+x2)^5$ を展開したときの $x0^2\ x1^2\ x2^1$ の個数は,

多項定理より $5!/(2!\times 2!\times 1!)$ です．
$(x0+x1+x2)^5$ $(0,\ 1,\ k$はすべて異なる「xの添え数字」$)$ の個数は 1 なので，
$T[5]$ の $x0^2\ x1^2\ x2^1$ の個数は $5!/(2!\times 2!\times 1!) = \binom{5}{2,\ 2,\ 1}$ です．

以上と $x_0 \sim x_6$ の対称性より， $T[5]$ は次と等しいです．

$$T[5] = 15\left(x0^5 + x1^5 + \cdots + x6^5\right) + 5 * \binom{5}{2}\left(x0^2\,x1^3 + x0^3\,x1^2 + \cdots + x5^2\,x6^3 + x5^3\,x6^2\right) +$$
$$5 * \binom{5}{1}\left(x0^1\,x1^4 + x0^4\,x1^1 + \cdots + x5^1\,x6^4 + x5^4\,x6^1\right) +$$
$$\binom{5}{2,\ 2,\ 1}\left(x0^2\ x1^2\ x2^1 + x0^2\ x1^1\ x2^2 + \cdots + x4^2\ x5^2\ x6^1 + x4^2\ x5^1\ x6^2\right) +$$
$$\binom{5}{3,\ 1,\ 1}\left(x0^3\ x1^1\ x2^1 + x0^1\ x1^3\ x2^1 + \cdots + x4^1\ x5^3\ x6^1 + x4^1\ x5^1\ x6^3\right)$$

即ち

$$T[5] = 15\,S_5 + 5 * \binom{5}{2} S_{2,3} + 5 * \binom{5}{1} S_{1,4} + \binom{5}{2,\ 2,\ 1} S_{2,2,1} + \binom{5}{3,\ 1,\ 1} S_{3,1,1}$$

一般に $m = 1,\ 2,\ 3..$ に対し
$T[m] =$

$$\sum_{x,y,z\text{はfの異なる解}} (x+y+z)^m = 15\,S_m + 5 \sum_{i\ge j,\,i+j=m} \binom{m}{i} S_{i,j} + \sum_{i\ge j\ge k,\,i+j+k=m} \binom{m}{i,\ j,\ k} S_{i,j,k}$$

§2-5 T[m]の値から A_i を求める

定義より $T[m]$ は F_{35} の異なる解のm乗の総和と一致します．
故にニュートン恒等式の $i \le n$ の式で「$S_i \to T[i]$， $a_i \to A_i$」と入れ替えて，

$$A_0 = 1,\quad -i\,A_i = T[1]\,A_{i-1} + T[2]\,A_{i-2} + \cdots + T[i-1]\,A_1 + T[i]\quad (1 \le i \le 35)$$

$T[1]$，$T[2]$， ..， $T[35]$ の値は求まるので， A_i を $A_1 \sim A_{i-1}$ で表すことができて，
A_i が求まります． 結局 §2 − 2 では 方程式の係数から解のk乗の総和を求めたのに対し，
ここでは逆に解のk乗の総和から方程式の係数を求めます． 上手いです．

【コメント】§2 − 2 の【コメント】の例で言うと (カ) ~ (エ) より「S_1 が求まれば a_1」，
「S_1，S_2，a_1 から a_2」，「S_1，S_2，S_3，a_1，a_2 から a_3」が求まります．
この様にして$n = 3$ の時は S_1，S_2，S_3 から a_1，a_2，a_3 が求まります．

§3 Mathematica による S_i, $S_{i,j}$, $S_{i,j,k}$,T[m]の計算

§3 – 1 S_i の計算

以上の式とMathematica を使って「 $S_i \to S_{i,j} \to S_{i,j,k} \to T[m] \to A_i \to$ F35」
の順に求めていきます. まず $S_i = S[i]$ は ニュートン恒等式より,
次の様に計算されます. F_{35} を計算するためなので S_i ($1 \le i \le 35$) を求めます.
この計算は数分 (私の環境で約8分) かかるので結果のみを closed cell に入れてあります.

```
as = {a1, a2, a3, a4, a5, a6, a7};
S[1] = -a1;
S[i_] := S[i] = If[i <= 7, Simplify[-Sum[as[[k]]  S[i - k], {k, 1, i - 1}] - i  as[[i]]], Simplify[-
Sum[as[[k]]  S[i - k], {k, 1, 7}]]]
Stable = Table[S[k], {k, 1, 35}]
```

(結果は↓の中 . 評価してください)

In[3293]:=

```
S[i_] := Stable[[i]]
```

§3 – 2 $S_{i,j}$, $S_{i,j,k}$, T[m] の計算

$S_{i,j}$, $S_{i,j,k}$, T[m] ($1 \le i, j, m \le 35$) については,
漸化式ではなく, S_i ($1 \le i \le 35$) の値から直接計算できます.

In[3294]:=

```
S[{i_, j_}] := If[i ≠ j, S[i] × S[j] - S[i + j], (S[i]^2 - S[2 i]) / 2] // Simplify
S[{i_, j_, k_}] := Which[i == j == k, (S[i] × S[{i, i}] - S[{2 i, i}]) / 3, i == j && j ≠ k,
   S[k] × S[{i, i}] - S[{i + k, i}], i ≠ j && j == k, S[i] × S[{k, k}] - S[{i + k, k}],
   (i - j) (j - k) (i - k) ≠ 0, S[i] × S[{j, k}] - S[{i + j, k}] - S[{i + k, j}]] // Simplify
```

T[m] の定義のために「 $S2[\{i, j\}] = \binom{i+j}{i} S[i, j]$,

$S3[\{i, j, k\}] = \left(\begin{matrix} i+j+k \\ i, j, k \end{matrix}\right) S[i, j, k]$ 」と定義します.

In[3296]:=

```
S2[x_List] := Multinomial[x[[1]], x[[2]]] * S[x]
S3[x_List] := Multinomial[x[[1]], x[[2]], x[[3]]] * S[x]
```

これらを使って T[m] を定義します.

In[3298]:=

```
T[m_] := 15 S[m] + 5 Total[S2 /@ IntegerPartitions[m, {2}]] +
   Total[S3 /@ IntegerPartitions[m, {3}]] // Simplify
```

少し例を見ておきます.

In[3299]:=

```
T[1]
```

Out[3299]=

$-15\ a1$

In[3300]:=

```
T[2]
```

Out[3300]=

$5\ (3\ a1^2 - 4\ a2)$

In[3301]:=

```
T[3]
```

Out[3301]=

$-15\ a1^3 + 30\ a1\ a2 - 6\ a3$

これらは，§1の$T[m] の結果を，基本対称式を使って変形した結果と一致します．

In[3302]:=

```
SymmetricReduction[#, vars, {-a1, a2, -a3, a4, -a5, a6, -a7}][[1]] & /@
 {$T[1], $T[2], $T[3]}
```

Out[3302]=

$\{-15\ a1,\ 15\ a1^2 - 20\ a2,\ -15\ a1^3 + 30\ a1\ a2 - 6\ a3\}$

§3 − 3　A_i（$1 \le i \le 35$）の計算とF_{35} の式

T[k]（$1 \le k \le 35$）の値から A_i（$1 \le i \le 35$）の値を求めるには，次の漸化式を使います．

In[3303]:=

```
A[0] = 1;
A[k_] := A[k] = -1 / k SymmetricReduction[Sum[T[i] × A[k - i], {i, 1, k}],
      vars, {-a1, a2, -a3, a4, -a5, a6, -a7}][[1]] // Simplify
```

最初に求めた A_1，A_2，A_3 の結果と比較すると，確かに一致しています．

In[3305]:=

```
A[1]
```

Out[3305]=

$15\ a1$

In[3306]:=

```
A[2]
```

Out[3306]=

$5\ (21\ a1^2 + 2\ a2)$

In[3307]:=

```
A[3]
```

Out[3307]=

$455\ a1^3 + 140\ a1\ a2 + 2\ a3$

以上からF_{35} は次の様に求まります．しかし計算には非常に時間が掛かるので 入力は text形式とし，
結果は下のclosed cell に入れてあります．

```
As=Table[A[k], {k, 1, 35}]
F35 = x^35 + Sum[As[[k]] x^(35 - k), {k, 1, 35}];
```

(F_{35}は ↓にあります． 非常に長いです．評価してください)

§3 – 4 例

- Gal = F_{14} =
  D_7 の方程式「$x^7 + 4x^4 + x^3 - 2x^2 + 2x + 3 = 0$」に対して f_{35} を求めて因数分解してみます.

In[3309]:=

```
f35 = F35 /. AssociationThread[{a1, a2, a3, a4, a5, a6, a7} → {0, 0, 4, 1, -2, 2, 3}]
```

Out[3309]=

$$31104 - 456192x + 2460672x^2 - 6314400x^3 + 8701608x^4 - 5329568x^5 - 1558016x^6 + 4343495x^7 + 1410916x^8 - 7779816x^9 + 4672350x^{10} + 3918452x^{11} - 3562668x^{12} - 1014192x^{13} + 1955754x^{14} + 493776x^{15} - 269874x^{16} + 571536x^{17} + 812340x^{18} + 333945x^{19} + 195750x^{20} + 110184x^{21} + 50280x^{22} - 14064x^{23} + 78x^{24} - 1692x^{25} + 684x^{26} - 258x^{27} + 834x^{28} + 32x^{29} + 20x^{30} - 8x^{31} + 8x^{32} + x^{35}$$

In[3310]:=

```
Factor[f35]
```

Out[3310]=

$$\left(27 - 54x + 18x^2 + 29x^3 + 2x^4 - 10x^5 + x^7\right)\left(-1 + 8x - 16x^2 + 15x^3 - 6x^4 + 2x^5 + x^7\right)\left(-9 + 42x + 2x^2 + 25x^3 + 2x^4 + 8x^5 + x^7\right)\left(128 + 160x^3 + 56x^4 + 352x^6 + 469x^7 + 176x^8 + 7x^{10} + 10x^{11} + x^{14}\right)$$

- Gal = F_{42} の方程式「$x^7 - 2x^5 + x^4 + 4x^3 - x^2 - 4x + 3 = 0$」
  に対して f_{35} を求めて因数分解してみます.

In[3311]:=

```
f35 = F35 /. AssociationThread[{a1, a2, a3, a4, a5, a6, a7} → {0, -2, 1, 4, -1, -4, 3}]
```

Out[3311]=

$$15111 - 276132x + 1426637x^2 - 3680789x^3 + 4781023x^4 - 863982x^5 - 6771888x^6 + 9142466x^7 - 937547x^8 - 8716581x^9 + 7475377x^{10} + 2428894x^{11} - 7106606x^{12} + 2272884x^{13} + 3476289x^{14} - 2909525x^{15} - 415490x^{16} + 1819025x^{17} - 422522x^{18} - 537798x^{19} + 420339x^{20} + 120094x^{21} - 137300x^{22} + 13676x^{23} + 44668x^{24} - 8463x^{25} - 9548x^{26} + 2175x^{27} + 978x^{28} - 643x^{29} - 26x^{30} + 148x^{31} + 2x^{32} - 20x^{33} + x^{35}$$

In[3312]:=

```
Factor[f35]
```

Out[3312]=

$$\left(207 - 102x - 67x^2 + 887x^3 + 473x^4 - 472x^5 + 15x^6 + 415x^7 + 65x^8 - 58x^9 + 19x^{10} + 6x^{11} - 8x^{12} + x^{14}\right)\left(73 - 1298x + 6276x^2 - 15422x^3 + 22924x^4 - 21630x^5 + 12825x^6 - 5247x^7 + 4299x^8 - 5985x^9 + 5329x^{10} - 2439x^{11} + 430x^{12} + 113x^{13} + 321x^{14} - 192x^{15} + 72x^{16} + 33x^{17} - 4x^{18} - 12x^{19} + x^{21}\right)$$